
A statement in a natural language contains three parts: subject, verb andAbstract Description:
object. A statement in a computer language contains only two parts: verb
and an object; the subject, or actor, is the computer’s operating system.
The predecessor of the meme list processing is the computer language
LISP, in which every statement is a function call and the object is a list. The
list processing paradigm of programming contains these steps: 1. identify
an object, an item; 2. write a function, process, or procedure for an item; 3.
prepare a list of items; 4. use a loop on the list, to process each item.

This paper reviews the author’s development of the concept of list process-
ing and its implementation in SAS R© software.

The purpose of this exposition is to highlight the author’s papers publishedPurpose:
on these topics and to provide a critique of earlier ideas. This is accom-
plished by a review of the development of processes for calculating the
cardinality ratios cardinality types of the variables in a data set.

programmersAudience:

arrays of macro variables; cardinality ratio; compiled or executed contents,Keywords:
extended; database vocabulary; data structure and algorithm; dosubl func-
tion; list processing; macro arrays; scl functions: open, close, attrn(nobs,
nvars), fetchobs, getvarc, getvarn, varname, vartype,

This paper uses font changes to identify SAS words in text, documentationstyle:
words in italics and notable key words in SMALL CAPS; e.g.: the libname

statement allocates a libref ; CARDINALITY TYPES are in FEW, MANY and
UNIQUE.

If you think of standardization as the best that you know today, but which isquote:
to be improved tomorrow — you get somewhere. – Henry Ford

Introduction

This is the overview, which consists of a list of topics in this section.Overview

• definition of a list
• learning a computer language
• arrays of macro variables
• database vocabulary

Definition of a list

This section contains these topics.Overview

• list processing definitions
• contents listing
• contents program
• contents output data set

List processing
definitions

list : contains items
item : can be an attribute, or a list

attribute : description, information, number, or string
CARDINALITY : of a list is the number of items in the list

same as dimension of an array, number of observations in data set
SAS list : is a CONTROL DATA SET, where each row contains list of values for

a process, and the variable names are parameter names of that pro-
cess; it is like a database DIMENSION TABLE, see page 7;

compare to proc format options cntlin, cntlout (control-in and -out)Notes:
and data structure provided by proc contents

proc contents data = sashelp.class;contents listing
The CONTENTS Procedure

Data Set Name SASHELP.CLASS Observations 19

Member Type DATA Variables 5

Label Student Data

Alphabetic List of Variables and Attributes

# Variable Type Len

- -------- ---- ---

3 Age Num 8

4 Height Num 8

1 Name Char 8

2 Sex Char 1

5 Weight Num 8

ATTRIBUTES of a data set include: libref (sashelp) is a reference to a folder;Notes:
data set name (class) is the name of a file in the folder; data set label
(Student Data) provides a description of the data set; dimensions of the
data matrix are height (Observations 19) and width (Variables 5); list of
the attributes of variables in the data set includes the ROW IDENTIFIERS,
variable number (labeled #) and name and type in (Char,Num).

contents program *name: make-list-names-contents-demo.sas;

%let data = sashelp.class;

PROC contents data = &data noprint

out = list_names

(keep = libname memname varnum nobs

name type length format label);

run; * updates sysnobs and syslast;

PROC sql; describe table &syslast;

quit;

proc print data = &syslast;

run;

contents output data
set

create table WORK.LIST_NAMES

LIBNAME char(8) label=’Library Name’,

MEMNAME char(32) label=’Library Member Name’,

NAME char(32) label=’Variable Name’,

TYPE num label=’Variable Type’, <---<<< num

LENGTH num label=’Variable Length’,

VARNUM num label=’Variable Number’,

LABEL char(256) label=’Variable Label’,

FORMAT char(32) label=’Variable Format’,

NOBS num label=’Observations in Data Set’

type is numeric which differs from listing of type in (1=Num,2=Char)Notes:

contents output data
set

Obs LIBNAME MEMNAME NAME TYPE LENGTH VARNUM LABEL FORMAT NOBS

--- ------- ------- ---- ---- ------ ------ ----- ------ ----

1 SASHELP CLASS Age 1 8 3 19

2 SASHELP CLASS Height 1 8 4 19

3 SASHELP CLASS Name 2 8 1 19

4 SASHELP CLASS Sex 2 1 2 19

5 SASHELP CLASS Weight 1 8 5 19

Type in in (1=Num,2=Char), see standarization on page 7Notes:

Learning a computer language

• variables local or global type (boolean, char, num)
• conditions if then... else add code, or branch
• loops conditional exit while, leave, until

enumerate (start, stop, step)
itemize

• functions macro, method, process, procedure
• setup for startup configuration, autoexec, libraries
• compile or execute? data structure or algorithm?

Q: What is the difference between process and procedure?Notes:

process : is a set of numbered steps which are always performed;
testing is simple because output is consistent

procedure : is a set of choices which may add extra statements
— e.g., options — or branch:
if type eq char then ... else if type eq num then ...;
testing is complex due to multiple paths through the code

• learning a computer language Statz [49], Heilmann and W3C [45]citations:
• conditions: using logical operators and, or, not [42]

sysfunc and ifc: [28] writing testing-aware programs: [22]
• loops [23]
• macro design [36]
• setup for startup: autoexec companion, [37],

and sysparm companion, [40]

Arrays of macro variables

This section contains these topics.Overview

• data array and loops
• warning when only single type
• subroutine: make list values sort
• make macro array: data, symputx
• log: array of global macro variables
• make macro array: sql
• demo macro %do loop
• programming issues
• citations

data array and loops Here is a generic method to write each row of a data set to the log.

*name: demo-data-array.sas;

%let lib_data = sashelp.class;

*let lib_data = sashelp.class(keep=age height weight);*type=n;

DATA _null_;

if 0 then set &lib_data; *read data structure;

array _c(*) $32 _character_; *fragile;

array _n(*) _numeric_ ; *fragile;

do _obs = 1 to _n_obs; *loop, enumerated;

set &lib_data nobs = _n_obs

point = _obs;

do _i = 1 to dim(_c);

putlog _c(_i)= @;

end;

do _i = 1 to dim(_n);

putlog _n(_i)= @;

end;

put;*CR/LF: newline;

end;

stop;

run;

The array statement is paired with the dim function.Notes:
The dimension of an array is the CARDINALITY of the array.
Compare to programs cx-include page 14
and demo-macro-function page 15.

warning when only
single type

The above program gives a warning when all variables are a single type.

2 %let lib_data = sashelp.class(keep=age height weight);*type=n;

3 DATA _null_;

4 if 0 then set &lib_data;

5 array _c(*) $32 _character_; *fragile;

WARNING: Defining an array with zero elements.

See how to eliminate this warning in program cx-include on page 14.Notes:

This program is used as a subroutine in the next two examples.subroutine: make list
values sort

*name: make-list-values-sort.sas;

PROC sort data = &lib_data (keep = &var)

nodupkey

out = list_values;

by &var;

run;

%put echo &=sysnobs;

make macro array:
data, symputx

This is a classic program which uses the row number, n , as a natural key.

*name: demo-make-macro-array-data.sas;

%let lib_data = sashelp.class;

%let var = sex;

%include ’make-list-values-sort.sas’;

DATA _null_;

set &syslast end = endofile nobs = dimension;

call symputx(catt(’value’,_n_),&var);

if endofile then call symputx("dim_values",dimension);

run;

%put _global_;

log: array of global
macro variables

This log shows the macro variables of the parameters lib data and var,
and the sequentially-numbered macro variables with prefix value and the
CARDINALITY of the macro array dim values.

GLOBAL LIB_DATA sashelp.class

GLOBAL VAR sex

GLOBAL VALUE1 F

GLOBAL VALUE2 M

GLOBAL DIM_VALUES 2

Arrays of macro variable can also be produced with the sql procedure.make macro array: sql

*name: demo-make-macro-array-sql.sas;

%let lib_data = sashelp.class;

%let var = sex;

%include ’make-list-values-sort.sas’;

PROC sql noprint;

select &var

into :value1 -

from &syslast;

quit;

%let dim_values = &sqlobs;

%put _global_;

run;

the high value may be also coded into :value1 - :value&sysmaxlongNotes:

demo macro %do loop Arrays of macro variables are referenced in a macro %do loop using the
double ampersand syntax.

*name: demo-echo-macro-array.sas;

%include ’demo-make-macro-array-data.sas’;

*include ’demo-make-macro-array-sql.sas’;

%macro echo_items();

%do i = 1 %to &dim_values;

%put value&i: &&value&i;

*code for each item;

%end;

%mend echo_items;

%echo_items()

The name of the macro array is hard-coded and is separate from the macroprogramming issues
%do loop.
There is no function for the dimension, it must be fetched.
Macro variables are created in the global symbol table.
Macro with %do loop reads the global symbol table.
Code inside loop cannot be tested as a unit.
All these issues are addressed in the section on loops, page 12.

• macro array using call symput [14]citations:
used in [12] and [13]

• macro array using proc sql select into [16]
• macro array as 1. function which returns dimension, and 2. procedure

so that calling macro can declare macro variables local [1]
• macro do loop [7]

Database vocabulary

database contains
columns

primary key : row identifier, UNIQUE NATURAL KEY has values 1–nobs
foreign keys : categories, FEW, link to primary key of dimension table

composite key : set of foreign keys
fact : continuous, MANY, measurement or quantity; summable
text : information about primary key

boolean : event success
date : and/or time: event granularity for start/end of periodic snapshot

database contains
tables

fact : is a record of events ex: transaction history, inventory
contains primary key, foreign keys, facts observation

dimension : is also known as a lookup table
contains primary key, text

snapshots : are called reports
periodic : e.g. monthly (grain) summary

accumulating : tasks of project, budget record with monthly sums

database vocabulary, [32] and Kimball and Ross [46]citations:

The three basic lists

Lists In programming we use the lists of variables (names), data sets (mem-
names), and values. These lists are produced with these processes.

names : contents, data with scl functions, sql dictionary.columns
memnames : contents, sql dictionary.tables

values : frequency, sort, sql, summary

• Other commonly-used lists are available from sql dictionaries; thesecitations:
include user- and system-defined global macro variables, catrefs, fil-
erefs, librefs, options, and running text in footnotes and titles.
How to use sql select into for list processing [29] code: [4]

• SAS software maintains other functions in catalogs; these include
user-defined formats and macro definitions.
SASautos companion, [18] lists macro catalogs, code: [3]

Variable names

Procedures contents and sql produce a list of variable names; scl functionsOverview
can be used in a data step. Each method differs in the values of type.
contents (1,2) scl (C,N) sql (char,num)
The programs shown below standardize type into (c,n).
This section contains these topics.

• list-names, data structure
• list-names, printed
• list names, contents
• list names, scl
• list names, sql

list-names, data
structure

create table work.list_names
libname char(8),
memname char(32),
varnum num,
name char(32),
type char(1),
length num,
format char(40),
label char(40)

list-names, printed libname memname varnum name type length format label
sashelp class 1 Name c 8
sashelp class 2 Sex c 1
sashelp class 3 Age n 8
sashelp class 4 Height n 8
sashelp class 5 Weight n 8

list-names, contents *name: make-list-names-contents-x.sas;

%put echo &libname..&memname;

PROC contents data = &libname..&memname noprint

out = list_names

(keep = libname memname varnum name nobs

type length format label );

run;

DATA &syslast (label = "&memname");

if 0 then do; * arrange data structure ;

set &syslast(keep = libname memname nobs varnum name);

attrib type length = $1;

set &syslast(keep = length format);

set &syslast(keep = label);

end;

retain _max_length_char 0;

drop _:; * _temp vars;

do until(endofile);

set &syslast (rename = (type = _type_n))

end = endofile;

if _type_n = 1 then type = ’n’;

else do; type = ’c’;

_max_length_char = max(_max_length_char,length);

end;

output;

end;

call symputx(’_max_length_char’,_max_length_char);

run;

%put echo &=_max_length_char;

PROC sort data = &syslast; *add attribute: sorted-by;

by name;

run;

%put trace make-list-names-contents-x ending;

list-names, scl *name: make-list-names-scl-x.sas;

%put echo &libname..&memname;

DATA list_names(label = "&memname");

attrib libname length = $ 8

memname length = $32

n_obs length = 8

varnum length = 8

name length = $32

type length = $ 1

length length = 8

format length = $40

label length = $40;

drop _:; ** temp vars;

retain libname "&libname" memname "&memname"

_max_length_char n_obs 0;

_dsid = open ("&libname..&memname");

_n_vars = attrn(_dsid,’nvar’);

n_obs = attrn(_dsid,’nobs’);

do varnum = 1 to _n_vars;

name = varname (_dsid,varnum) ;

type = lowcase(vartype (_dsid,varnum));

length = varlength(_dsid,varnum) ;

format = varfmt (_dsid,varnum) ;

label = varlabel (_dsid,varnum) ;

output;

if type eq ’c’ then

_max_length_char = max(_max_length_char,length);

end;

_rc = close(_dsid);

call symputx(’_max_length_char’,_max_length_char);

run;

%put echo &=_max_length_char;

%put trace make-list-names-scl-x ending;

list-names, sql *name: make-list-names-sql-x.sas note: does not have nobs;

%put echo &libname..&memname;

PROC sql noprint;

create table list_names (label = "&memname") as

select libname, memname, varnum, name, type length=1,

length, format, label

from dictionary.columns %*list-names;

where libname eq "%upcase(&libname)"

and memname eq "%upcase(&memname)"

and memtype eq ’DATA’;

select max(length) into :_max_length_char trimmed

from &syslast where type eq ’c’;

quit;

%put echo &=_max_length_char;

%put trace make-list-names-sql-x ending;

Code for the above programs and others is available on [2].citations:

Data set names: memnames

This list is called memnames because that is the variable name in the out-Overview
put data sets of both the contents and sql procedures.

• list-memnames, data structure
• list-memnames, printed
• list-memnames, contents
• list-memnames, sql
• programming issues

The programs shown below add memnum with values (1 : n obs), which is
the NATURAL KEY described on page 7.

list-memnames, data
structure

create table WORK.LIST_MEMNAMES

libname char(8) label=’Library Name’,

memnum num, label=’Member Number’,

memname char(32) label=’Member Name’,

nobs num label=’Number of Physical Observations’,

nvars num label=’Number of Variables’,

memlabel char(256) label=’Data Set Label’

list-memnames,
printed

libname memnum memname nobs nvars memlabel

SASHELP 1 AACOMP 1544 4

SASHELP 2 AARFM 61 4

SASHELP 3 ADSMSG 426 6

SASHELP 4 AFMSG 1090 6

SASHELP 5 AIR 144 2 airline data (monthly: JAN49-DEC60)

...

list-memnames,
contents

*name: make-list-memnames-contents.sas;

%put echo &=libname;

PROC contents data = &libname.._all_ noprint

out = list_memnames

(keep = libname memname nobs memlabel);

run;

%put echo &=sysnobs;

DATA &syslast (label = "&libname");

attrib libname length=$ 8 label=’library name’

memnum length= 8 label=’member number’

memname length=$ 32 label=’member name’

nobs length= 8 label=’number of observations’

nvars length= 8 label=’number of variables’

memlabel length=$256 label=’member label’;

retain libname "&libname" memnum nvars 0;

set &syslast;

nvars +1;

by memname;

if last.memname then do;

memnum +1;

output;

nvars = 0;

end;

run;

%put trace make-list-memnames-contents ending;

list-memnames, sql *name: make-list-memnames-sql.sas;

PROC sql noprint;

create table list_memnames(label = "&libname") as

select libname,

monotonic() as memnum label=’member number’,

memname, nobs, nvar as nvars, memlabel

from dictionary.tables

where libname eq "%upcase(&libname)"

and memtype eq ’DATA’;

quit;

%put trace make-list-memnames-sql ending;

The sql dictionaries of all librefs are refreshed before each procedure call;programming issues
the contents procedure reads only the libref specified and is therefore
faster.
This information is also available in a set of sashelp views; these views read
from sql dictionaries and are deprecated by the author for that reason.

Values
This section shows the macro proc-freq and the data structure. The pur-Overview
pose of this macro is produce a frequency of each variable, standardize the
data structure and stack the outputs into one data set.

This is the list of topics in this section.
• macro proc-freq
• list-values, data structure
• list-values, printed

macro proc-freq This macro is called by two programs, demo-call-macro-call-execute on
page 12 and demo-call-macro-sql-basic on page 13.
Compare to macro proc-freq-dosub on page 23.
%macro proc_freq(name,type);

%put echo &libname..&memname..&name &=type;

%let type = %lowcase(%substr(&type,1,1));

PROC freq data = &libname..&memname; *fragile: global;

tables &name / noprint

out = frequency

(rename = (&name = valu_&type));

run;

DATA &syslast;

attrib memname length = $32

name length = $32

valu_c length = $32 %*fragile;

/** valu_c length = $&_max_length_char /**robust*/

valu_n length = 8

count length = 8

percent length = 8

level length = 8;

retain memname "&memname" name "&name"

valu_c ’.’ valu_n . ;

set &syslast;

level = _n_; *natural key;

run;

PROC append data = &syslast

base = list_values;

run;

%put trace &sysmacroname(&=name) ending;

%mend proc_freq;

This macro is adapted from [20].citations:

list-values, data
structure

create table WORK.LIST_VALUES
libname char(8),
memname char(32),
varnum num,
name char(32),
valu_c char(??), <---<<<
valu_n num,
count num,
percent num,
leven num

list-values, printed This report shows cardinality of each variable on its last line <---<<<.
Obs memname name valu_c valu_n count percent level

1 class Age 11.0 2 10.5263 1

2 class Age 12.0 5 26.3158 2

[snipped]

5 class Age 15.0 4 21.0526 5

6 class Age 16.0 1 5.2632 6 <---<<<

7 class Height 51.3 1 5.2632 1

8 class Height 56.3 1 5.2632 2

[snipped]

22 class Height 69.0 1 5.2632 16

23 class Height 72.0 1 5.2632 17 <---<<<

24 class Name Alfred . 1 5.2632 1

25 class Name Alice . 1 5.2632 2

[snipped]

41 class Name Thomas . 1 5.2632 18

42 class Name William . 1 5.2632 19 <---<<<

43 class Sex F . 9 47.3684 1

44 class Sex M . 10 52.6316 2 <---<<<

45 class Weight 50.5 1 5.2632 1

[snipped]

59 class Weight 150.0 1 5.2632 15 <---<<<

The length of variable valu c is marked as fragile; note that the programsprogramming issues
that produce the data set list-names shown above on 8 – 9 also provide
a macro variable max-length-char.

Loops: processes for lists

This section contains the specific routines that are the predecessors to the
Overview three general-purpose, reusable routines.

• specific solutions
– type macro calls, how to automate this?
– call execute macro
– sql select into

• general solutions, reusable
– cx-include: call execute an %include
– call-macro: call a macro
– call-text: create and expand references to macro variable in text

Both call-macro and call-text are macro functions; they do not return SASNotes:
statements, only macro language text.

type the macro calls In the beginning, cut and paste the list of variable names from the contents

output and type in the constant text of the macro calls.

*name: demo-typed-list-values.sas;

/** use list from proc contents printed output;

Alphabetic List of Variables and Attributes

# Variable Type Len

3 Age Num 8

4 Height Num 8

1 Name Char 8

2 Sex Char 1

5 Weight Num 8

/***********************/

%let libname = sashelp;

%let memname = class;

%proc_freq(Age ,Num )

%proc_freq(Height,Num )

%proc_freq(Name ,Char)

%proc_freq(Sex ,Char)

%proc_freq(Weight,Num )

PROC print;

run;

Specific Routines

call execute macro
calls

The call execute routine has been around since SAS.v6. This program
shows how to call a macro with two parameters.
*name: demo-call-macro-call-exec.sas;

%let libname = sashelp;

%let memname = class;

%include ’make-list-names-contents-x.sas’;

DATA _null_;

do until(endofile);

set list_names end = endofile;

*call exec of a macro is fragile without nrstr;

call execute(catt(’%proc_freq(name=’,name,’,type=’,type,’)’));

end;

stop;

run;

This solution may fail, not because of the code in this program, but whenNotes:
the macro definition contains what I call complexity: it contains any of these
statements: %do, %if, call symput, or symputx.

The error is difficult to diagnose because the macro will pass a unit test and
this program will work with simple macros but not with complex ones.

An explanation of the timing issues is in Fehd and Carpenter [44].

The solution with %nrstr is shown in program cx-include on page 14.

sql select into The sql dictionary.columns eliminates the separate step of making the
list of variable names.

*name: demo-call-macro-sql-basic.sas;

%let libname = sashelp;

%let memname = class;

PROC sql noprint;

select catt(’%proc_freq(’ %*fragile length= 200;

,name,’,’,type,’)’)

into :list %*fragile length=65534;

separated by ’ ’

from dictionary.columns %*list-names;

where libname eq "%upcase(&libname)"

and memname eq "%upcase(&memname)"

and memtype eq ’DATA’;

quit;

&list

%symdel list;

run;

PROC print;

run;

Ah, yes, this is my most spectacular fail! I published this in the fall of 2007Notes:
[21], and one week later got an e-mail from a user complaining that it broke
on a data set with three thousand (3,000) variables!

It may fail for either of two reasons.

cat function : the default length of text returned by any of the cat functions is $200;
the short length results in truncation of the macro call, which has no
closing parenthesis. This can be fixed by adding a length clause after
the function call: catt(...) length = $nnnn.

macro variable : the maximum length of text in a macro variable is (216)− 2 = 65, 534;
a possible solution is to reduce the name of the macro definition to
a single character: %macro z(...) otherwise, this error cannot be
fixed; consider one of the robust solutions, shown next.
A tutorial on sql is available in [29].citations:

Call execute an %include

This is the main algorithm of CallXproc.sas from the SmryEachVar suiteOverview
of sgf.2008 [24]. This program was published later as CallXinc.sas [27].

Compare to the demonstration program demo-data-array.sas on page 5.

Note that the length of the variable stmt is marked as fragile. The value is
max(for i=1 to n-vars (length(%let var(i)=value(i);))

,length(%include &cx include;))

where var(i) is length(name(i)) and value(i) is length(i).

Robust code for cx-include is available as callxinc on [26] which has acitations:
data step to calculate the maximum length for stmt.

cx-include *name: cx-include.sas;

%put echo &=cx_data &=cx_include;

DATA _null_;

if 0 then set &cx_data;

attrib _stmnt length = $128 %*fragile;

_name length = $ 32

_i length = 4;

array _var_c(*) _character_ ; *robust: _name;

array _var_n(*) _numeric_ ; *robust: _i;

do until(endofile);

set &cx_data end = endofile;

do _i = 1 to dim(_var_c) -2; *note: dim()-2;

_name = vname(_var_c(_i));

_stmnt = catx(’ ’,’%let’,_name,’=’,_var_c(_i));

link cx_stmnt;

end;

do _i = 1 to dim(_var_n) -1; *note: dim()-1;

_name = vname(_var_n(_i));

_stmnt = catx(’ ’,’%let’,_name,’=’,_var_n(_i));

link cx_stmnt;

end;

_stmnt = "%include &cx_include";

link cx_stmnt;

end;

stop;

cx_stmnt: call execute(cats(’%nrstr(’,_stmnt,’;)’));

return;

run;

This program hides the complexity of the call execute(...%nrstr()) inNotes:
the data step subroutine labeled cx stmnt which is called with the link

statement.
Example usage of this program is shown in
make-list-cr-types-n-steps on page 19 and
make-list-cr-types-1-step-libname on page 25.

Demo macro function which reads a data set

This demonstration macro shows the common code in macros call macroOverview
and call text.
Compare to programs demo-data-array.sas on page 5. and cx-include

on page 14.
This program uses scl functions like make-list-names-scl-x on page 8.

This functions calls either of the functions getvarc or getvarn with theNotes:
reference getvar&type. This trick eliminates the conditional statements
%if &type eq c %then getvarc(); %else getvarn().

demo macro function %MACRO demo_macro_function

(data = sashelp.class);

%local dsid n_obs n_vars obs rc name type value;

%let dsid = %sysfunc(open (&data ));

%let n_obs = %sysfunc(attrn(&dsid,nobs ));

%let n_vars = %sysfunc(attrn(&dsid,nvars));

%do obs = 1 %to &n_obs;

%let rc = %sysfunc( fetchobs(&dsid,&obs));

%do varnum = 1 %to &n_vars;

%let name = %sysfunc( varname(&dsid,&varnum));

%let type = %sysfunc( vartype(&dsid,&varnum));

%let value = %sysfunc(getvar&type(&dsid,&varnum));

%put echo &name = &value;

%end;

%put;

%end;

%let rc = %sysfunc(close(&dsid));

%mend demo_macro_function;

log echo Name = Alfred

echo Sex = M

echo Age = 14

echo Height = 69

echo Weight = 112.5

echo Name = Alice

echo Sex = F

echo Age = 13

echo Height = 56.5

echo Weight = 84

Call-macro

The purpose of macro call macro is to assemble a macro call of the form:Overview
%macro-name(var(i)=value(column(i),row(j)),...)

The %do loops for obs and varnum and the fetch of name and type are
exactly the same as shown above in demo-macro-function.

call-macro %MACRO call_macro

(data = sashelp.class /* required */

,macro_name = put); /* default for testing */

%let dsid = %sysfunc(open (&data ));

%let n_obs = %sysfunc(attrn(&dsid,nobs ));

%let n_vars = %sysfunc(attrn(&dsid,nvars));

%do obs = 1 %to &n_obs;

%let list_parameters =;

%let rc = %sysfunc(fetchobs(&dsid,&obs));

%do varnum = 1 %to &n_vars;

%let name = %sysfunc(varname (&dsid,&varnum));

%let type = %sysfunc(vartype (&dsid,&varnum));

%* add name=value to list;

%let list_parameters=&list_parameters&name=%left(

%sysfunc(getvar&type(&dsid,&varnum)));

%* add comma to end of parameter list;

%if &varnum lt &n_vars %then

%let list_parameters = &list_parameters,;

%end;

%&macro_name(&list_parameters);%*semicolon not required!;

%end;

%let rc = %sysfunc(close(&dsid));

%mend call_macro;

Macro callmacr is described in [34], code is available on [5].citations:

Call-text

The purpose of macro call text is to assemble macro assignment state-Overview
ments and then expand &text which contains references to &var1--&varn.
%let var1=value1;...;%let varn=valuen;%unquote(&text)

call-text %MACRO call_text

(data = sashelp.class

,text = %nrstr(%put echo: &=name sex=&sex;)

,global = 0); /* make mvars global? */

%let _global = %eval(not(0 eq &global));

%let _text = &text;

%let _dsid = %sysfunc(open (&data ));

%let _n_obs = %sysfunc(attrn(&_dsid,nobs ));

%let _n_vars = %sysfunc(attrn(&_dsid,nvars));

%do _obs = 1 %to &_n_obs;

%let _rc = %sysfunc(fetchobs (&_dsid,&_obs ));

%do _varnum = 1 %to &_n_vars;

%let _name = %sysfunc(varname (&_dsid,&_varnum));

%let _type = %sysfunc(vartype (&_dsid,&_varnum));

%if &_global %then %global &_name;

%else %local &_name;

%let &_name= %sysfunc(getvar&_type(&_dsid,&_varnum));

%*** note functions: getvarc and getvarn;

%end;

%unquote(&_text)

%end;

%let _rc = %sysfunc(close(&_dsid));

%mend call_text;

Macros call-macro and call-text fail when text variables such as labelsprogramming issues
contain special characters of the macro language, ampersand (&), percent
(%), or function delimiter, comma (,), or statement delimiter, semicolon (;),
or unmatched (single or double) quotes or parentheses.

The solution is to drop those variables, data=x(drop=label).

Macro calltext is described in [41], code is available on [6].citations:

Code for macros call-macro and call-text contains tests for the exis-
tence of the data set; tests for existence of various SAS objects is shown
in [43] code: [9]

Macro dateloop is a function which returns a series of dates and surround-
ing text; it is described in [31] code: [8]

A function to calculate cardinality ratios and types

The cardinality of a set is the number of elements in the set. The cardi-Overview
nality of a data set is the number of rows in the data set. The cardinality
of a variable is the number of distinct values within the variable; in SAS
this is called n-levels and is available from the frequency procedure. The
CARDINALITY RATIO (CR) of a variable is the cardinality divided by the data
set nobs. The CARDINALITY TYPE (cr-type) is determined by comparing the
CR with the mean of all the CR in the data set.

This section contains these programs.

• N steps page 19
• Two steps with freq.nlevels and scl functions page 20
• One step with scl and dosubl page 23
• Cardinality ratio types of all memnames in a libref page 25

This overview contains these topics.

• cr-types, data structure
• cr-types, printed
• algorithm

cr-types, data
structure

create table WORK.CR_TYPES(label=’class nobs=19 nvars=5’)

memname char(32),

varnum num,

cr_type char(10),

card_ratio num format=BESTD7.5,

n_levels num format=BEST8. label=’Number of Levels’,

name char(32) label=’Table Variable’,

type char(1),

length num,

format char(49),

label char(256)

cardinality types,
printed

card_ n_

memname varnum cr_type ratio levels name type length format label

------- ------ ------- ------- ------ ------ ---- ------ ------ -----

class 1 .unique 1 19 Name c 8

class 2 few 0.10526 2 Sex c 1

class 3 few 0.31579 6 Age n 8

class 4 many 0.89474 17 Height n 8

class 5 many 0.78947 15 Weight n 8

algorithm 1. fetch n-obs: nobs(data set) or from list-memnames
2. for each variable, fetch n-levels
3. cardinality-ratio = n-levels / n-obs
4. calculate mean of cardinality ratios
5. cardinality-ratio-type = compare cardinality ratio to mean

The first paper to identify cardinality ratio was [25]; other papers includecitations:
[32] and [33]. Macro cr-calc, shown in [35], is the basis for the n-steps
programs shown next. This paper also includes programs for the mode and
summary procedures.

N steps

Implementation of this algorithm requires 7 + n-vars steps.Overview
This is the list of topics in this section.

• make-list-cr-types-n-steps
• proc-freq-out-name
• make-list-card-ratios-from-work-name
• proc-smry-cr
• make-list-types-from-cr

make-list-cr-types-n-
steps

This program may also use the macro calltext for the loop.
*name: make-list-cr-types-n-steps.sas;

%let libname = sashelp;

%let memname = class;

%include ’make-list-names-contents-x.sas’; *steps=2;

/**** loop: freq of each variable **************************/

%let cx_data = &syslast;

%let cx_include = ’proc-freq-out-name.sas’; *steps=nvars;

%include ’cx-include.sas’; *steps=1;

/************************************************************

*call_text(data = &syslast

,text = %nrstr(%include ’proc-freq-out-name.sas’;)

,global = 1)/**************************************/

%include ’make-list-card-ratios-from-work-name.sas’; *steps=1;

%include ’proc-smry-cr.sas’; *steps=1;

%include ’make-list-types-from-cr.sas’; *steps=1;

PROC print; *steps=1;

title3 "Cardinality Ratios and Types: &libname..&memname";

run;

proc-freq-out-
name.sas

*name: proc-freq-out-name.sas;

PROC freq data = &libname..&memname;

tables &name / noprint

out = freq_&name; * fragile: length=$32;

run;

Compare to programs proc-freq page 11 and proc-freq-dosub page 23.

make-list-card-ratios-
from-work-name.sas

*name: make-list-card-ratios-from-work-name.sas;

PROC contents data = work._all_ noprint

out = list_n_levels

(keep = memname name nobs label

where =( memname like ’FREQ_%’

and label eq ’ ’)

rename =(nobs = n_levels));

run;

PROC sort data = &syslast;

by name;

DATA list_card_ratios;

do until(endofile);

merge list_names &syslast (keep = name n_levels)

end = endofile;

by name;

card_ratio = n_levels / nobs;

output;

end;

stop;

run;

proc-smry-cr.sas *name: proc-smry-cr.sas;

PROC summary data = list_card_ratios;

output out = _mean_cr (drop = _freq_ _type_)

mean(card_ratio)= _mean_cr;

run;

make-list-types-from-
cr.sas

*name: make-list-types-from-cr.sas;

DATA list_cr_types;

drop _:;

set _mean_cr;

putlog ’echo’ _mean_cr=;

do until(endofile);

set list_card_ratios end = endofile;

select;

when(n_levels eq 1) cr_type = ’n-levels=1’;

when(card_ratio eq 1) cr_type = ’.unique’;

when(card_ratio gt _mean_cr) cr_type = ’many’;

otherwise cr_type = ’few’;

end;

output;

end;

stop;

run;

This algorithm contains many steps which are interdependent; this makesprogramming issues
unit testing a sequential process.

Two steps with freq.nlevels and scl functions

This section contains these topics.Overview

• fetch n-obs, n-vars
• fetch n-levels
• describe output data structure
• calculate cardinality ratios and mean
• assign cr-type

fetch n-obs, n-vars %let _dsid = %sysfunc(open (&libname..&memname,i));

%let _n_obs = %sysfunc(attrn(&_dsid,nobs));

%let _n_vars = %sysfunc(attrn(&_dsid,nvar));

%let _rc = %sysfunc(close(&_dsid));

fetch n-levels ODS select none; * noprint;

PROC freq data = &libname..&memname nlevels;

ods output

nlevels= list_names

(keep = tablevar nlevels

rename=(tablevar= name nlevels= n_levels));

run;

ODS select all;

describe data
structure

DATA &syslast;

attrib memname length = $32

varnum length = 8

cr_type length = $ %length(n-levels=1)

card_ratio length = 8 %*range=(0:1];

format = bestd7.5

n_levels length = 8

name length = $32

type length = $ 1

length length = 8

format length = $49

label length =$256;

calculate cardinality
ratios and mean

array _cr(&_n_vars);

** loop: for each row, calculate cardinality ratio;

do _i = 1 to dim(_cr);

set &syslast (keep = n_levels) point = _i;

_cr(_i) = n_levels / &_n_obs; %*global mvar;

end;

***** calculate mean for select card_ratio to cr_type;

_mean_cr = mean(of _cr(*));

fetch variable
information

_dsid = open("&libname..&memname");

do _i = 1 to dim(_cr);

set &syslast point = _i;

*...;

card_ratio = _cr(_i);

The assumption is data and freq.n-levels are in varnum order.Notes:

assign cr-type select;

when(n_levels eq 1 ) cr_type = ’n-levels=1’;

when(card_ratio eq 1 ) cr_type = ’.unique’;

when(card_ratio gt _mean_cr) cr_type = ’many’;

otherwise cr_type = ’few’;

end;

output;

There are a number of functions for arrays, this paper [38] shows callcitations:
sortn.

Previous papers on cardinality ratio include [32] [33], [35]

make-list-cr-types-2-
steps

*name: make-list-cr-types-2-steps.sas;

*called by: make-list-cr-types-2-steps-memname.sas

ODS select none; * noprint;

PROC freq data = &libname..&memname nlevels;

ods output

nlevels= list_names

(keep = tablevar nlevels

rename=(tablevar= name nlevels= n_levels));

run;

ODS select all;

%let _dsid = %sysfunc(open (&libname..&memname,i));

%let _n_obs = %sysfunc(attrn(&_dsid,nobs));

%let _n_vars = %sysfunc(attrn(&_dsid,nvar));

%let _rc = %sysfunc(close(&_dsid));

DATA cr_types(label = "&memname nobs=&_n_obs nvars=&_n_vars");

attrib memname length = $32

varnum length = 8

cr_type length = $ %length(n-levels=1)

card_ratio length = 8 %*range=(0:1];

format = bestd7.5

n_levels length = 8

name length = $32

type length = $ 1

length length = 8

format length = $49

label length =$256;

array _cr(&_n_vars);

drop _:; * _temp vars;

retain memname "&memname";

** loop: for each row, calculate cardinality ratio;

do _i = 1 to dim(_cr);

set &syslast (keep = n_levels) point = _i;

_cr(_i) = n_levels / &_n_obs; *global macro variable;

end;

***** calculate mean for select card_ratio to cr_type;

_mean_cr = mean(of _cr(*));

_dsid = open("&libname..&memname");

do _i = 1 to dim(_cr);

set &syslast point = _i;

varnum = varnum (_dsid,name );

type = lowcase(vartype (_dsid,varnum));

length = varlength(_dsid,varnum);

format = varfmt (_dsid,varnum);

label = varlabel (_dsid,varnum);

card_ratio = _cr(_i);

select;

when(n_levels eq 1 ) cr_type = ’n-levels=1’;

when(card_ratio eq 1 ) cr_type = ’.unique’;

when(card_ratio gt _mean_cr) cr_type = ’many’;

otherwise cr_type = ’few’;

end;

output;

end;

_rc = close(_dsid);

stop;

run;

The freq.nlevels data set is output in varnum order so i is varnum.Notes:

One step with scl functions and dosubl

This is the list of topics in this section.Overview

• make list cr types for memname
• proc freq dosub
• programming issues
• make list cr types 1 step
• calculate card-ratios and mean
• assign cr-type

make list cardinality
types for memname

*name: make-list-cr-types-1-step-memname.sas;

* see also make-list-cr-types-1-step-libname.sas;

%let libname = sashelp;

%let memname = class;

%let memnum = 0;

%let out_lib = library;

%let out_lib = work;

%include ’make-list-cr-types-1-step.sas’;

PROC print data = &syslast noobs;

title3 "Cardinality Ratios, Types of &libname..&memname";

run;

proc freq dosub %macro proc_freq_dosub

(data = &libname..&memname /*global*/

,name =

,type =

,varnum =

,memnum = 0

,out_lib = work );

%put echo &sysmacroname &=name &=type &=varnum &=memnum;

PROC freq data = &data;

tables &name / noprint

out = &out_lib..freq_&memnum._&varnum

(label = "&name"

rename = (&name = valu_&type));

run;

%put echo &sysmacroname &=sysnobs;

%global _n_levels;

*robust: avoid divide by zero in calling program;

*** sysnobs returns -1 for data-set w/nobs=0;

%if &sysnobs gt 0 %then

%let _n_levels = &sysnobs;

%else %let _n_levels = 1;

%mend proc_freq_dosub;

In the first edition of this macro I assumed that the global system macroprogramming issues
variable sysnobs would be updated. As you can see in the log the state-
ment %put echo &sysmacroname &=sysnobs; prints the number of
observations of the data set that the procedure produced. However, in pro-
gram make-list-cr-types-1-step the fetch of sysnobs fails.
nlvl(varnum)=input(symget(’sysnobs’),32.);

Thus the solution shown here: allocate a global macro variable n-levels

and conditionally set its value to sysnobs.
Function dosubl is described in Langston [47].citations:

make list cr types 1
step

* name : make-list-cr-types-1-step.sas;

* called by: make-list-cr-types-1-step-memname.sas;

* called by: make-list-cr-types-1-step-libname.sas;

%put echo &libname..&memname &=memnum &=out_lib;

%let _dsid = %sysfunc(open (&libname..&memname));

%let _n_obs = %sysfunc(attrn(&_dsid,nobs));

%let _n_vars = %sysfunc(attrn(&_dsid,nvar));

%let _rc = %sysfunc(close(&_dsid));

DATA &out_lib..list_cr_types_&memnum

(label = "memname=&memname,obs=&_n_obs,vars=&_n_vars");

attrib memname length = $32

memnum length = 8 label = ’mem num’

varnum length = 8 label = ’var num’

cr_type length = $ %length(n-levels=1)

label = ’card. ratio type’

card_ratio length = 8 label = ’card. ratio’

format = bestd7.5

n_levels length = 8 label = ’n-levels’

name length = $32

type length = $ 1

length length = 8

format length = $49

label length =$256;

array _cr (&_n_vars) ;

array _name (&_n_vars) $32;

array _nlvl (&_n_vars) ;

array _type (&_n_vars) $ 1;

drop _:;

retain memname "&memname" memnum &memnum;

_dsid = open("&libname..&memname");

do varnum = 1 to dim(_name);

_name(varnum)= varname(_dsid,varnum) ;

_type(varnum)= lowcase(vartype(_dsid,varnum));

_rc = dosubl(catt(’%proc_freq_dosub(name=’,_name(varnum)

,’,type=’,_type(varnum)

,’,varnum=’ ,varnum

,",memnum=&memnum,out_lib=&out_lib)"));

_nlvl(varnum) = input(symget(’_n_levels’),32.);

if _nlvl(varnum) and &_n_obs then

_cr(varnum) = _nlvl(varnum)/&_n_obs;

else _cr(varnum) = 1/&sysmaxlong;*small and non-zero;

end;

***** calculate mean for select card_ratio to cr_type;

_mean_cr = mean(of _cr(*));

do varnum = 1 to dim(_name);

name = _name ( varnum);

card_ratio = _cr ( varnum);

n_levels = _nlvl ( varnum);

type = _type ( varnum);

length = varlength(_dsid,varnum);

format = varfmt (_dsid,varnum);

label = varlabel (_dsid,varnum);

select;

when(n_levels eq 1 ) cr_type = ’n-levels=1’;

when(card_ratio eq 1 ) cr_type = ’.unique’;

when(card_ratio gt _mean_cr) cr_type = ’many’;

otherwise cr_type = ’few’;

end;

output;

end;

_rc = close(_dsid);

run;

Cardinality types of all memnames in a libref

Notice that program make-list-cr-types-1-step-memname on page 23Overview
has a parameter memnum.
Routine make-list-cr-types-1-step on page 24 creates a report with a
unique name make-list-cr-types-* and passes this on to
macro proc-freq-dosub on page 23 which creates a unique name for each
output data set freq-(memnum)-(varnum) from the frequency procedure.

This section has these topics.

• make list cr types for libname
• list of data sets

make list cr types for
libname

* name: make-list-cr-types-1-step-libname.sas;

* see also make-list-cr-types-1-step-memname.sas;

%let libname = sashelp; *17K files! time= 5+ hours!;

%let libname = library;

*let libname = my_lib;

%let out_lib = library;

%let out_lib = work;

%include ’make-list-memnames-contents.sas’;

%let cx_data = &syslast(keep = nobs memname memnum

where =(nobs));

%let cx_include = ’make-list-cr-types-1-step.sas’/source2;

%include ’cx-include.sas’/nosource2;

PROC sql; select memname, nobs, nvar, memlabel length=32

from dictionary.tables

where libname eq "%upcase(&out_lib)"

and memtype eq ’DATA’;

quit;

run;

list of data sets Number of Number

Physical of

Member Name Obs Vars Data Set Label

-------------------- --------- ------ ------------------------------

FREQ_1_1 9 3 locale

FREQ_1_2 184 3 key

FREQ_1_3 1 3 lineno

FREQ_1_4 1345 3 text

FREQ_2_1 1 3 locale

...

LIST_CR_TYPES_1 4 11 memname=AACOMP,obs=1544,vars=4

LIST_CR_TYPES_2 4 11 memname=AARFM,obs=61,vars=4

...

Programs in this paper are available on [10].citations:

Summary
A SAS data set is an example of a list, the attribute items of which —n-Conclusion
obs and n-vars — contain information that can be used to process each
variable and provide additional attributes of that variable — n-levels — that
can be compared not only to the other variables but also to the data set
set attribute, n-obs. The new attributes of cardinality ratio and cardinality
type explain the three major categories of variables: row-identifier: unique,
classification: few, and analysis: many.
This information can then be used to coordinate other programming tasks.

This paper describes a set of tools which can be used to analyze a data set or allfuture work
data sets in a libref. Here are my plans for future work.

• make list of user-defined formats and convert data set of frequency of variable
to frequency of variable with format, [15]

• for card-type equal many and type equal numeric,
– find macro freqall in [20] and modify it into macro freq hilo; produce

a report of all the high and low values of each numeric
– write a macro smry from freq where the summary procedure reads the

frequency output data set with weight count;

• for few: calculate max-length-char and write a report to stack (set) or append
into one data set all the categorical variables

• all(few) x each(many): make a list of the cr-type eq ’few’ variables and
create a program to do a cross-tabulation and summary of each cr-type eq
’many’ variables, where type eq numeric per algorithm of Raithel [48]

• exclude cr-type of UNIQUE and N-LEVELS=1 from calculation of mean(cr)

The SAS community wikipedia has pages for the programs shown here.program downloads
See also the category macros by Ron Fehd.

• macro arrays [11] 1998, v1, call symputx
[17] 2004, v2, sql select into [1] 2007, v3, a function and procedure

• making lists contains make-list-memnames, make-list-names and
ml-name-x:
make-list-names-cardinality-ratio-2-steps, page 22

• loops cx-include, formerly callxinc 2008 and cxinclude 2012
call-macro call-text

• cardinality ratio database vocabulary summarize each var
• program design [19], [22], [39] and [30]citations:

Quentin McMullen and Rick Langston reviewed the next-to-last draft andAcknowledgements
provided helpful commentary. Art Carpenter shared the idea of a list as a
control data set.

